

Synthèse : Puissance 4 (en F)

Corrigé

1 puissance4

Le jeu « Puissance 4 » est proposé par MB Jeux. Sur la boîte, on peut lire les indications suivantes.

Puissance 4 est un jeu de stratégie verticale passionnant et plein d’astuce. C’est un jeu facile à apprendre et amusant à jouer. Ses règles sont simples. Chaque joueur essaie de faire une rangée de quatre pions dans le cadre – horizontalement, verticalement ou diagonalement – tout en essayant d’empêcher son adversaire de faire de même. Croyez-moi ce n’est pas aussi facile que cela en a l’air ! La position verticale du jeu demande au joueur beaucoup de concentration et de réflexion.

Montage :

1. Suivez les instructions illustrées pour l’assemblage des parties en plastiques (non reproduites ici).
2. Placez le jeu entre les joueurs.
3. Chaque joueur choisit une couleur de pions.

But du jeu :

Être le premier joueur à placer quatre de ses pions sur une ligne horizontale, verticale ou diagonale continue.

Règles du jeu :

1. Choisissez le premier joueur. Le joueur qui commence la première partie sera le deuxième joueur au cours de la deuxième partie.
2. À tour de rôle, chaque joueur fait tomber un de ses pions dans une des fentes au sommet de la grille.
3. Le jeu continue jusqu’à ce qu’un des joueurs ait un alignement continu de quatre pions de sa couleur. L’alignement peut être vertical, horizontal ou en diagonale.
4. Pour vider la grille, poussez la barre de retenue qui se trouve au vase de celle-ci et les pions tomberont. Vous pouvez maintenant commencer la partie suivante.

Conseil : Lire attentivement les questions suivantes et ébaucher une solution au brouillon avant de commencer la rédaction.

Exercice 1 : Préparation du jeu

Le jeu « puissance 4 » se joue sur une grille verticale de six lignes et sept colonnes avec 21 pions rouges et 21 pions jaunes.

Pour faciliter les calculs ultérieurs, nous allons ajouter une bordure à cette grille. Ses dimensions sont donc de huit lignes et neuf colonnes. Bien entendu, ces cases supplémentaires ne peuvent pas contenir de pions (qu'ils soient jaunes ou rouges). Elles sont donc toujours vides.

Définir les types nécessaires pour représenter la grille de jeu.

```

1  ! Définir le type pour une case de la grille
2
3  INTEGER, PARAMETER, PUBLIC :: VIDE=0
4  INTEGER, PARAMETER, PUBLIC :: ROUGE=1
5  INTEGER, PARAMETER, PUBLIC :: JAUNE=2
6
7  ! La grille de jeu avec une bordure supplémentaire
8  TYPE, PRIVATE :: typ_grille
9      INTEGER, DIMENSION(0:NB_LIGNES+1,0:NB_COLONNES+1)::grille
10 END TYPE typ_grille

```

Exercice 2 : Vider la grille

Pour commencer chaque partie, il faut commencer par vider la grille des jetons qu'elle contient. Écrire un sous-programme qui réalise cette opération.

Solution : Ce sous-programme a un seul paramètre, la grille, en **out**. Il s'agit donc d'une initialisation. Nous en faisons donc une procédure.

```

1  Procédure vider(grille: out Grille) Est
2      -- vider la grille.

```

Ce sous-programme consiste à mettre toutes les cases de la grille à **VIDE**, y compris les cases de la bordure. On le fait donc naturellement avec deux boucles imbriquées, l'une qui balaie les lignes, l'autre les colonnes.

```

1  ! Vider la grille. Ceci correspond à actionner la barre se situant
2  ! sous la grille du jeu réel.
3  !
4  ! Assure :
5  !      -- la grille est vide
6
7  SUBROUTINE vider(grille)
8      INTEGER, DIMENSION(:,:), INTENT(OUT)::grille
9      grille(:,:) = VIDE
10
11 END SUBROUTINE vider

```

Exercice 3 : Afficher la grille

Écrire un sous-programme qui affiche le contenu de la grille. Par convention, les jetons de couleur rouge seront représentés par un astérisque (*) et les jetons jaunes par la lettre o minuscule.

Pour faciliter, la localisation d'une case de la grille, les numéros de lignes et de colonnes seront affichés comme sur la figure 1.

```

      1 2 3 4 5 6 7
6     *           6
5     o           5
4     o o *       4
3     * o   o o   3
2     * o *   o *  2
1 * o * * o o *  1
      1 2 3 4 5 6 7

```

FIGURE 1 – Affichage de la grille

Solution : Ce sous-programme prend un seul paramètre, la grille. C'est un paramètre en entrée. De plus ce sous-programme fait de l'affichage. Nous en faisons donc une procédure.

```

1 Procédure afficher(grille: in Grille) Est
2   -- Afficher la grille.

```

L'implémentation de ce sous-programme n'est pas très facile à structurer, en particulier si on veut essayer de limiter les redondances qui rendent plus difficiles les évolutions ultérieures.

Comme on ne sait pas se positionner sur l'écran, il faut donc afficher ligne par ligne. On se rend compte rapidement qu'il nous faudra donc faire deux boucles imbriquées : l'une pour les lignes, l'autre pour les colonnes. On peut réaliser une boucle qui s'appuie sur l'affichage à l'écran et qui inclut donc la bordure correspondant aux numéros. Il suffit alors de tester les valeurs des indices de boucle des lignes et colonnes pour savoir si :

- on n'affiche rien (les quatre coins) ;
- on affiche un numéro de ligne ou de colonne (les bordures) ;
- on affiche le contenu d'une case (les autres cases).

Une autre solution, celle retenue ici, consiste à séparer la première et la dernière ligne qui consiste à afficher les numéros de colonne. Notons que pour afficher les lignes, on commence par celle de numéro NB_LIGNES car les indices de notre tableau correspondent aux numéros de ligne du puissance 4 (on aurait pu prendre la convention inverse).

```

1 R1 : Raffinage De « afficher »
2   | Afficher les numéros de colonnes
3   | Pour i <- NB_LIGNES Décrémenter Jusqu'À i = 1_LIGNES Faire
4   |   | Afficher la ligne i
5   | FinPour
6   | Afficher les numéros de colonnes
7
8 R2 : Raffinage De « Afficher la ligne i »
9   | Afficher le numéro de ligne
10  | Pour j <- 1 Jusqu'À j = NB_COLONNES Faire
11  |   | afficher_case(grille[i][j])
12  | FinPour
13  | Afficher le numéro de ligne

```

Comme l'affichage des numéros de colonne est à faire en début et fin, nous en faisons un sous-programme pour factoriser le code.

```

1  ! Afficher les numéros de colonnes
2  SUBROUTINE afficher_numeros_colonnes()
3      INTEGER:: j          ! parcourir les colonnes de la grille
4      PRINT"(A3,7I2)", "___", (j,j=1,NB_COLONNES)
5  END SUBROUTINE afficher_numeros_colonnes
6
7  ! Afficher la grille et son contenu
8  SUBROUTINE afficher(grille)
9      INTEGER, DIMENSION(:, :), INTENT(IN):: grille
10     INTEGER:: i, j       ! parcourir les cases de la grille
11     CALL afficher_numeros_colonnes()
12
13     DO i = NB_LIGNES, 1, -1
14         ! afficher le numéro de ligne à gauche
15         ! afficher la ligne i
16         ! afficher le numéro de ligne à droite
17         PRINT"(I3,7A2,I3)", i, (symbole_case(grille(i,j)), j=1,NB_COLONNES), i
18         PRINT*
19     END DO
20
21     CALL afficher_numeros_colonnes()
22 END SUBROUTINE afficher

```

Exercice 4 : Décider si un coup est possible

Écrire un sous-programme qui indique s'il est possible de jouer dans une colonne donnée.

Exemple : Sur l'exemple de la figure 1, il est possible de jouer les colonnes 1, 2, 3, 5, 6 ou 7. Il n'est pas possible de jouer la colonne 4.

Solution : Ce sous-programme prend en paramètre :

- la grille (**in**);
- le numéro de colonne envisagé (**in**);
- l'information de si le coup est possible (**out**).

Un seul paramètre en **out**, d'autres paramètres en **in**. On en fait donc une fonction.

```

1  Fonction est_coup_possible(grille: in Grille; colonne: in Entier): Booléen Est
2      -- Est-ce qu'il est possible de jouer dans la colonne donnée
3      -- de la grille ?

```

Pour savoir s'il est possible de jouer, il suffit de regarder si la colonne est pleine ou non, c'est-à-dire s'il sa dernière case, celle du haut, est vide ou non. Si elle est vide, on peut jouer dans la colonne. Si elle n'est pas vide, on ne peut pas y jouer. Notons que si elle est vide, ça ne veut pas dire que le pion restera dans cette case. Il tombera...

On en déduit donc naturellement le code.

```

1  ! est-ce que la colonne peut être jouée sur cette grille ?
2  FUNCTION est_coup_possible(grille, colonne) RESULT(resultat)
3      INTEGER, DIMENSION(:, :), INTENT(IN):: grille
4      INTEGER, INTENT(IN):: colonne
5      LOGICAL:: resultat

```

```

6   resultat=(grille(NB_LIGNES,colonne) == VIDE)
7   END FUNCTION est_coup_possible

```

Exercice 5 : Lâcher un jeton

Lorsqu'un joueur joue, il lâche un jeton de sa couleur au dessus d'une colonne de la grille. En raison de la force gravitationnelle, le jeton tombe dans la grille jusqu'à ce qu'il soit arrêté par un autre jeton ou le fond de la grille.

Écrire un sous-programme qui réalise cette opération : il calcule le nouvel état de la grille lorsqu'un jeton est lâché au dessus d'une colonne donnée.

Exemple : Partant de la figure 1, si les rouges jouent la colonne 5, le nouvel état de la grille est celui donné par la figure 2.

	1	2	3	4	5	6	7	
6				*				6
5				o				5
4			o	o		*		4
3			*	o		o	o	3
2		*	o	*	*	o	*	2
1	*	o	*	*	o	o	*	1
	1	2	3	4	5	6	7	

FIGURE 2 – Les rouges ont joué la colonne 5

Solution : Le sous-programme prend en paramètre :

- la grille (**in out**);
- la colonne donnée (**in**);
- le jeton, ou plutôt sa couleur (**in**).

Comme il y a un paramètre en **in out**, nous en faisons une procédure.

```

1   Procédure lâcher(grille: in out Grille;
2       colonne: in Entier;
3       couleur: in Case) Est
4       -- La couleur joue un pion dans la colonne.
5       --
6       -- Nécessite
7       -- est_coup_possible(grille, colonne)

```

Le principe de l'algorithme est alors de regarder jusqu'où va tomber le jeton, soit il n'y a encore aucun pion et il est donc en fond de grille, soit il est dans la dernière case qui est vide. En fait, le plus simple est de partir du fond de la colonne et de chercher la première case vide en remontant. Comme on sait qu'il est possible de jouer dans la colonne (précondition), on n'a pas besoin de tester le débordement de la colonne.

```

1   R1 : Raffinage De « lâcher »
2       | Déterminer la ligne de chute
3       | Mettre le pion dans cette case
4
5   R2 : Raffinage De « Déterminer la ligne de chute »

```

```

6   | ligne <- 1
7   | TantQue grille[colonne, ligne] <> VIDE Faire
8   |   | ligne <- ligne + 1
9   | FinTQ

```

Remarque : On décide de faire un sous-programme, en l'occurrence une fonction, de « Déterminer la ligne de chute » car nous aurons par la suite l'utilité de trouver cette ligne de chute. Il n'est pas évident ici, du premier coup, de voir qu'il peut être utile d'en faire un sous-programme.

```

1  ! ligne sur laquelle tombe un jeton lâché à la colonne spécifiée
2  !
3  ! Nécessite
4  !     est_coup_possible(grille, colonne)      -- coup possible
5
6  FUNCTION ligne_de_chute(grille, colonne)RESULT(resultat)
7      INTEGER, DIMENSION(:, :), INTENT(IN)::grille
8      INTEGER, INTENT(IN)::colonne
9      INTEGER::resultat
10
11     INTEGER:: l           ! parcourir les lignes de la colonne considérée
12
13     !     assert(est_coup_possible(grille, colonne))
14
15     DO l=1,NB_LIGNES
16         IF(grille(l,colonne) == VIDE)THEN
17             resultat=l
18             EXIT
19         ENDIF
20     ENDDO
21 END FUNCTION ligne_de_chute
22
23 ! Lacher sur la grille à la colonne précisée un jeton de la couleur
24 ! donnée.
25 !
26 ! Nécessite
27 !     est_coup_possible(grille, colonne)      -- coup possible
28 !
29 ! Assure
30 !     -- un pion de plus sur la colonne
31 !     -- le nouveau pion est de la couleur donnée
32
33 SUBROUTINE lacher(grille,colonne,couleur)
34     INTEGER, INTENT(IN)::couleur
35     INTEGER, INTENT(IN)::colonne
36     INTEGER, DIMENSION(:, :), INTENT(OUT)::grille
37     !     assert(est_coup_possible(grille, colonne))
38
39     grille(ligne_de_chute(grille, colonne),colonne) = couleur
40 END SUBROUTINE lacher

```

Exercice 6 : Compter les jetons alignés

Écrire un sous-programme qui indique quelle est la longueur de l'alignement le plus long qui in-

clut un jeton donné (repéré par son numéro de colonne et son numéro de ligne). Les alignements doivent être cherchés dans toutes les directions (horizontale, verticale ou en diagonale).

Ce sous-programme sera ensuite utilisé pour déterminer la fin d'une partie (alignement ≥ 4) ou aider l'ordinateur à choisir la colonne où jouer (exercice 8).

Sur l'exemple de la figure 2, la case (1,1) correspond à un alignement de trois jetons en diagonale ; la case (7,1) correspond à un alignement de deux jetons verticalement ; la case (6,2) correspond à trois jetons alignés verticalement ou en diagonale...

Indication : Quel est l'intérêt d'avoir défini une bordure supplémentaire autour de la grille ?

Solution : Ce sous-programme prend en paramètre une grille (**in**), une ligne (**in**) et une colonne (**in**) et détermine le plus long alignement (**out**). Un paramètre en sortie. Tous les autres en entrée. Nous pouvons donc en faire une fonction.

```

1  Fonction nb_pions_alignés(grille: in Grille;
2      ligne: in Entier;
3      colonne: in Entier): Entier
4      -- Plus grand nombre de jetons de la même couleur alignés
5      -- horizontalement, verticalement ou en diagonale et incluant
6      -- le jeton de la position (colonne, ligne).
7      --
8      -- Nécessite
9      -- grille[colonne, ligne] <> VIDE

```

L'algorithme est un peu compliqué, ou tout au moins fastidieux. Il faut regarder dans les quatre directions (horizontalement, verticalement et en diagonale) pour compter l'alignement le plus grand. Pour chaque direction, il faut compter la case (colonne, ligne) et les deux sens de la direction.

Supposons que l'on sache calculer un nombre de pions alignés selon une direction. On peut alors en déduire le code de « nb_pions_alignés »

```

1  R1 : Raffinage De « nb_pions_alignés »
2  | Résultat <- max(nb_pions_dir(grille, colonne, ligne, 1, 1),
3  |                 max(nb_pions_dir(grille, colonne, ligne, 1, 0),
4  |                 max(nb_pions_dir(grille, colonne, ligne, 1, -1),
5  |                 nb_pions_dir(grille, colonne, ligne, 0, 1)))

```

Il ne reste plus qu'à écrire nb_pions_dir. C'est une fonction.

```

1  Fonction nb_pions_dir(grille: in Grille; colonne, ligne: in Entier;
2      dir_x, dir_y: in Entier) Est
3      -- Compter le nombre de pions dans la direction (dir_x, dir_y)
4      -- et incluant le pion en (colonne, ligne).
5      --
6      -- Nécessite
7      -- grille[colonne, ligne] <> VIDE
8      -- (dir_x <> 0) Ou (dir_y <> 0)    -- direction valide
9      --
10     -- Assure
11     -- Résultat >= 1                -- au moins le jeton de (colonne, ligne)

```

Pour son implémentation, il suffit de compter le jeton de (colonne, ligne) puis compter les jetons de même couleur dans la direction (dir_x, dir_y) et ceux dans la direction opposée

```

(-dir_x, -dir_y).
1  R1 : Raffinage De « nb_pions_dir »
2  | couleur <- grille[colonne, ligne]
3  | Résultat <- 1
4  | Comptabiliser les jetons dans la direction (dir_x, dir_y)
5  | Comptabiliser les jetons dans la direction (-dir_x, -dir_y)

1  ! Plus grand nombre de jetons de la même couleur alignés suivant la
2  ! direction (dir_x, dir_y) et incluant le jeton de la case
3  ! (ligne,colonne).
4  !
5  ! Nécessite
6  !     Grille(ligne)(colonne) != VIDE           -- un jeton en (ligne, colonne)
7  !     dir_x != 0 || dir_y != 0                 -- une vraie direction
8  !
9  ! Assure
10 !     Résultat >= 1
11
12 FUNCTION nb_pions_dir(grille,ligne,colonne,dir_x,dir_y)RESULT(resultat)
13     INTEGER::resultat
14     INTEGER, DIMENSION(:,:),INTENT(IN)::grille
15     INTEGER, INTENT(IN)::colonne
16     INTEGER, INTENT(IN)::ligne
17     INTEGER, INTENT(IN)::dir_x
18     INTEGER, INTENT(IN)::dir_y
19     INTEGER:: l           ! ligne dans la direction +/- dir_y
20     INTEGER:: c           ! colonne dans la direction +/- dir_x
21     INTEGER:: couleur     ! la couleur de l'alignement
22
23 !     assert(grille(ligne)(colonne) != VIDE)
24 !     assert(dir_x != 0 || dir_y != 0)
25
26     couleur = grille(ligne,colonne)
27     resultat = 1           ! le jeton en (ligne, colonne)
28
29
30     ! Comptabiliser les jetons dans la direction (dir_x, dir_y)
31     l = ligne + dir_y
32     c = colonne + dir_x
33     DO
34         IF (grille(l,c) /= couleur)EXIT
35         resultat=resultat+1
36         l = l + dir_y
37         c = c + dir_x
38     END DO
39
40     ! Comptabiliser les jetons dans la direction (-dir_x, -dir_y)
41     l = ligne - dir_y
42     c = colonne - dir_x
43     DO
44         IF (grille(l,c) /= couleur)EXIT
45         resultat=resultat+1

```



```

46         l = l - dir_y
47         c = c - dir_x
48     END DO
49 END FUNCTION nb_pions_dir
50
51 ! Plus grand nombre de jetons de la même couleur alignés
52 ! horizontalement, verticalement ou en diagonale et incluant le jeton
53 ! de la position (ligne, colonne)
54 !
55 ! Assure
56 !     Résultat >= 0
57
58 FUNCTION nb_pions_alignes(grille,ligne,colonne) RESULT(resultat)
59     INTEGER:: resultat
60     INTEGER,DIMENSION(:,:),INTENT(IN)::grille
61     INTEGER,INTENT(IN)::colonne
62     INTEGER,INTENT(IN)::ligne
63
64     IF (grille(ligne,colonne) == VIDE)THEN
65         resultat = 0
66     ELSE
67         ! Déterminer le plus grand alignement
68         resultat =max(nb_pions_dir(grille, ligne, colonne, 1, 1),&
69             nb_pions_dir(grille, ligne, colonne, 1, 0),&
70             nb_pions_dir(grille, ligne, colonne, 1, -1),&
71             nb_pions_dir(grille, ligne, colonne, 0, 1))
72     END IF
73
74 END FUNCTION nb_pions_alignes

```

Exercice 7 : Conseiller une colonne où jouer

Écrire un sous-programme qui étant donné une grille propose une colonne où jouer. Le numéro de la colonne sera choisi aléatoirement.

Solution : Ce sous-programme est une fonction car il a un seul paramètre en sortie (la colonne conseillée) et un paramètre en entrée (la grille de jeu), éventuellement deux si l'on considère aussi la couleur de celui que l'on doit conseiller (couleur qui n'est pas exploité dans cette version très naïve du conseil).

```

1  Function colonne_conseillée_aléatoire(grille: in Grille; couleur: in Case) Est
2      -- colonne conseillée pour jouer

```

Le principe est de tirer une colonne au hasard jusqu'à ce qu'elle correspondent à une colonne où l'on peut jouer.

```

1  ! colonne conseillée pour jouer.
2  !
3  ! Principe : La colonne conseillée est choisie au hasard entre les
4  ! colonnes possibles.
5  !
6  ! Nécessite
7  !     -- la partie n'est pas finie
8  !

```

```

 9  ! Assure
10  !      est_coup_possible(grille, Résultat)
11
12  SUBROUTINE colonne_conseillee_aleatoire(grille,colonne)
13      INTEGER,DIMENSION(:,:),INTENT(IN)::grille
14      INTEGER,INTENT(OUT)::colonne
15      REAL::alea
16      DO
17          CALL RANDOM_NUMBER(alea)
18          colonne = alea* NB_COLONNES + 1
19          IF(est_coup_possible(grille, colonne))EXIT
20      ENDDO
21
22  END SUBROUTINE colonne_conseillee_aleatoire

```

Exercice 8 : Améliorer le conseil

Pour améliorer le conseil, on ajoute une contrainte supplémentaire : conseiller la colonne qui permet l'alignement le plus long.

Remarque : Cette contrainte implique que si un coup gagnant existe, la colonne conseillée conduira à la victoire.

Exemple : Partant de la figure 2, les jaunes joueront en colonne 5. Ceci provoque un alignement de 4 pions et donc la victoire.

Solution : Les paramètres sont les mêmes que pour le conseil précédent sauf que la couleur du joueur est ici nécessaire.

```

 1  Fonction colonne_conseillee(grille: in Grille; couleur: in Case) Est
 2      -- colonne conseillée pour jouer

```

Le principe du conseil est un peu plus compliqué car il faut proposer le coup correspondant à l'alignement le plus grand. On a déjà écrit un sous-programme pour calculer un alignement mais il suppose que l'on ait déjà joué dans la case. Il nous faut donc jouer dans la case. Et si l'on joue, il faut également pouvoir annuler le coup pour ne pas fausser la partie.

```

 1  R1 : Raffinage De « colonne_conseillee »
 2      | max <- 0          -- alignement maximal trouvé
 3      | Pour c <- 1 Jusqu'À c = NB_COLONNES Faire
 4      | | Jouer en c pour couleur
 5      | | Si alignement > max Alors
 6      | | | max <- alignement
 7      | | | Résultat <- c
 8      | | FinSi
 9      | | Annuler le coup en c
10     | FinPour

 1  ! Colonne conseillée pour jouer.
 2  !
 3  ! Principe : La colonne conseillée est celle qui permet l'alignement
 4  ! le plus long. Ainsi, si un coup est gagnant, la colonne proposée
 5  ! est gagnante.
 6  !
 7  ! Nécessite

```

```

 8 !           -- la partie n'est pas finie
 9 !
10 ! Assure
11 !           est_coup_possible(grille, Résultat)
12 !           -- si un coup gagnant existe, alors Résultat est gagnant
13
14 SUBROUTINE colonne_conseilee(grille,couleur,colonne)
15     INTEGER,DIMENSION(:,,:),INTENT(IN)::grille
16     INTEGER,INTENT(IN)::couleur
17     INTEGER,INTENT(OUT)::colonne
18     INTEGER:: c           ! pour chercher une colonne gagnante
19     INTEGER:: nb_pions     ! Nombre de pions alignés pour c
20     INTEGER:: max         ! le nb max de pions alignés (donc de resultat)
21     INTEGER:: ligne       ! conserver la ligne de chute du pion lâché
22     REAL::alea
23     INTEGER,DIMENSION(SIZE(grille,1),size(grille,2))::grille1
24     grille1=grille
25     ! chercher une colonne permettant le plus long alignement
26     max = 0               ! nécessairement un minorant strict
27     DO c = 1 ,NB_COLONNES
28         IF (est_coup_possible(grille, c))THEN
29             ! déterminer la ligne de chute
30             ligne = ligne_de_chute(grille, c)
31
32             ! jouer le coup pour voir
33             CALL lacher(grille1, c, couleur)
34
35             ! évaluer le coup
36             nb_pions = nb_pions_alignes(grille1, ligne, c)
37             CALL RANDOM_NUMBER(alea)
38             ! choisir le coup (par rapport au précédent)
39             IF (nb_pions > max.OR.&                !{ meilleur coup }
40                 (nb_pions == max &                !{ coup équivalent }
41                  .AND. alea>0.5))THEN            !{ un peu d'aléatoire }
42                 max = nb_pions
43                 colonne = c
44             ENDIF
45
46         ENDIF
47         grille1=grille
48     END DO
49 END SUBROUTINE colonne_conseilee

```